1-co middle c program.

The format is E = {Const,Linear,Quad,Pert} which means $O(\text{Const}\, e^{\text{Linear+Quad}}(1+\epsilon\,\text{Pert}))$ no tricky scalings.

This is a stand alone notebook but, all formulas were checked in NewVdVAlgebraAt15.nb

```
(*Can be simplified a bit more*)
Logos[α_, β_, δ_, μ_ , a_, c_, w_] :=
 -1/(4 μ^4) (1 + t) (4 a c w δ^2 μ^2 + δ (1 + μ) (w^2 α^2 + a^2 β^2) +
    a^2 w^2 δ^3 (1 + 3 μ) + (1 - t) (2 (α β + δ μ)^2 - α^2 β^2) +
   2 (α β + 2 δ μ + a w δ^2 (1 + 2 μ) + 2 c δ μ^2) (w α + a β) +
    4 (c μ^2 + a w δ (1 + μ)) (α β + δ μ) )
```

```
(*R-matrix*)
Rp[i_, j_] := {1, b c_j, a_i w_j, a_i c_i w_j + c_i c_j + a_i^2 w_j^2/4}
Rm[i_, j_] := {1, -b c_j, -t^-1 a_i w_j, -c_i c_j + t^-1 a_i c_j w_j - t^-2 a_i^2 w_j^2/4}
```

```
(*Cuaps*)
nr[i_] := {t^(1/2), 0, 0, -c_i}
nl[i_] := {1, 0, 0, 0}
ur[i_] := {t^(-1/2), 0, 0, c_i}
ul[i_] := {1, 0, 0, 0}
```

```
DP_{x_→D_α_, y_→D_β_}[P_][f_] :=
 Total[CoefficientRules[P, {x, y}] /. ({m_, n_} → c_) :> c D[f, {α, m}, {β, n}]]
```

```
(*Stitching operation*)
NO[w_i_, a_j_, k_][Z_] := Block[
  {α, β, δ, C, L, Q, P, QRest, μ, q },
  C = Z[[1]]; (*The (inverse of the) constant in front, aka ω*)
  L = Z[[2]]; (*The linear term of the exponential, unscaled*)
  Q = Z[[3]]; (*The quadratic term, unscaled.*)
  P = Z[[4]];
  q = μ^-1 ((1 - t) α β + α w_i + β a_j + δ a_j w_i);
  P = DP_{w_i→D_α, a_j→D_β}[P][Exp[q]] Exp[-q]; (*First deal with P formally*)
  (*Then set the variables*)
  α  = D[Q, w_i] /. a_j → 0;
  β  = D[Q, a_j] /. w_i → 0;
  δ  = D[Q, a_j, w_i];
  QRest = Q /. {a_j → 0, w_i → 0};
  μ  = 1 - (1 - t) δ;
```

```
  C = μ C;
  Q = QRest + μ^-1 ((1 - t) α β + α w_i + β a_j + δ a_j w_i);
  P = P + Logos[α, β, δ, μ , a_j, c_k, w_i];
  {C, L, Q, P} /. {b → Log[t]}
 ]

NO[w_k_, c_j_][Z_] := Block[
  {α, γ, C, L, Q, P, QRest, q },
  C = Z[[1]]; (*The (inverse of the) constant in front, aka ω*)
  L = Z[[2]]; (*The linear term of the exponential, unscaled*)
  Q = Z[[3]]; (*The quadratic term, unscaled.*)
  P = Z[[4]];
  q = ⅇ^γ α w_k + γ c_j;
  P = DP_{w_k→D_α, c_j→D_γ}[P][Exp[q]] Exp[-q]; (*First deal with P formally*)
  (*Then set the variables*)
  α  = D[Q, w_k];
  γ  = D[L, c_j];
  QRest = Q /. {w_k  → 0};

  Q = QRest + ⅇ^γ α w_k;
  {C, L, Q, P} /. {b → Log[t]}
 ]

NO[c_i_, a_k_][Z_] := Block[
  {β, γ, C, L, Q, P, QRest, q },
  C = Z[[1]]; (*The (inverse of the) constant in front, aka ω*)
  L = Z[[2]]; (*The linear term of the exponential, unscaled*)
  Q = Z[[3]]; (*The quadratic term, unscaled.*)
  P = Z[[4]];
  q = ⅇ^γ β a_k + γ c_i;
  P = DP_{a_k→D_β, c_i→D_γ}[P][Exp[q]] Exp[-q]; (*First deal with P formally*)
  (*Then set the variables*)
  β  = D[Q, a_k];
  γ  = D[L, c_i];

  QRest = Q /. {a_k  → 0};

  Q = Q /. {a_k -> ⅇ^γ a_k};
  (*Q=QRest+ⅇ^γβ a_k;*)
  {C, L, Q, P} /. {b → Log[t]}
 ]

m[i_, j_, k_][Z_] := Block[{x}, (Z // NO[w_i, a_j, x] // NO[c_i, a_j] // NO[w_i, c_j]) /.
    {a_{i|j} → a_k, c_{i|j|x} → c_k, w_{i|j} → w_k} // Together]
```

```
ZZ = {10 + t, -2 b c_1 + b c_2 + 3 b c_3, a_2 w_2 + a_3 w_2 + a_2 w_1 + a_1 w_1 + a_3 w_1 + a_1 w_2, 3 t a_1 w_1 + a_2 c_3 + t c_1};
```

```
(*Meta-associativity*)
(ZZ // m[1, 2, 1] // m[1, 3, 1] // Expand) -
  (ZZ // m[2, 3, 2] // m[1, 2, 1] // Expand) // Together
```

{0, 0, 0, 0}

```
(*Disjoint union of two tangles, multiplies the constant terms, adds the exponents*)
DisjUnion[L_] :=
 {Times @@ L[[All, 1]], Plus @@ L[[All, 2]], Plus @@ L[[All, 3]], Plus @@ L[[All, 4]]}
```

```
(*Reidemeister 2*)
DisjUnion@{Rp[1, 2], Rm[3, 4]} // m[1, 3, x] // m[2, 4, y]
```

{1, 0, 0, 0}

```
(*Reidemeister 3*)
(DisjUnion@{Rp[1, 2], Rp[3, 4], Rp[5, 6]} // m[1, 3, x] // m[2, 5, y] // m[4, 6, z]) -
  (DisjUnion@{Rp[1, 2], Rp[3, 4], Rp[5, 6]} // m[3, 5, x] // m[1, 6, y] // m[2, 4, z]) //
 Expand
(DisjUnion@{Rm[1, 2], Rm[3, 4], Rm[5, 6]} // m[1, 3, x] // m[2, 5, y] // m[4, 6, z]) -
  (DisjUnion@{Rm[1, 2], Rm[3, 4], Rm[5, 6]} // m[3, 5, x] // m[1, 6, y] // m[2, 4, z]) //
 Expand
```

{0, 0, 0, 0}

{0, 0, 0, 0}

```
(*Kinks*)
```

$Kp[i\_] := \left\{\sqrt{t},\ Log[t]\ c_i,\ a_i\ w_i,\ -c_i + c_i^2 + a_i\ c_i\ w_i + \frac{1}{4} a_i^2\ w_i^2\right\}$

$Km[i\_] := \left\{\frac{1}{\sqrt{t}},\ -Log[t]\ c_i,\ -\frac{a_i\ w_i}{t},\ c_i - c_i^2 + \frac{a_i\ c_i\ w_i}{t} - \frac{a_i^2\ w_i^2}{4\ t^2}\right\}$

```
DisjUnion[{Rp[1, 4], nr[2], ul[3]}] // m[1, 2, 1] // m[1, 3, 1] // m[1, 4, 1] // Together //
 Expand
DisjUnion[{Rm[4, 1], nr[2], ul[3]}] // m[1, 2, 1] // m[1, 3, 1] // m[1, 4, 1] // Together //
 Expand
DisjUnion[{Rp[4, 1], nl[2], ur[3]}] // m[1, 2, 1] // m[1, 3, 1] // m[1, 4, 1] // Expand
DisjUnion[{Rm[1, 4], nl[2], ur[3]}] // m[1, 2, 1] // m[1, 3, 1] // m[1, 4, 1] // Expand
```

$\left\{\sqrt{t},\ Log[t]\ c_1,\ a_1\ w_1,\ -c_1 + c_1^2 + a_1\ c_1\ w_1 + \frac{1}{4} a_1^2\ w_1^2\right\}$

$\left\{\frac{1}{\sqrt{t}},\ -Log[t]\ c_1,\ -\frac{a_1\ w_1}{t},\ c_1 - c_1^2 + \frac{a_1\ c_1\ w_1}{t} - \frac{a_1^2\ w_1^2}{4\ t^2}\right\}$

$\left\{\sqrt{t},\ Log[t]\ c_1,\ a_1\ w_1,\ -c_1 + c_1^2 + a_1\ c_1\ w_1 + \frac{1}{4} a_1^2\ w_1^2\right\}$

$\left\{\frac{1}{\sqrt{t}},\ -Log[t]\ c_1,\ -\frac{a_1\ w_1}{t},\ c_1 - c_1^2 + \frac{a_1\ c_1\ w_1}{t} - \frac{a_1^2\ w_1^2}{4\ t^2}\right\}$

```
(*3_1 0-Framed*)
Z = DisjUnion@{Rp[5, 1], Rp[2, 6], Rp[7, 3], ur[4], Km[8], Km[9], Km[10]};
For[i = 2, i ≤ 10, i++, Z = Z // m[1, i, 1]];
Z // Expand // Together
```

$$\left\{\frac{1-t+t^2}{t}, 0, 0, \left(-t+2t^2-3t^3+2t^4+2c_1-2tc_1+2t^3c_1-2t^4c_1-2a_1w_1-2t^3a_1w_1\right) \Big/ \left(1-t+t^2\right)^2\right\}$$

```
(*4_1 0-Framed*)
Z = DisjUnion@{Rm[8, 1], Rp[5, 9], Rm[2, 6], Rp[10, 3], nr[7], ur[4]};
For[i = 2, i ≤ 10, i++, Z = Z // m[1, i, 1]];
Z // Expand // Together
```

$$\left\{\frac{-1+3t-t^2}{t}, 0, 0, \frac{-1+t^2+2c_1-2t^2c_1-2a_1w_1-2ta_1w_1}{1-3t+t^2}\right\}$$

```
(*5_1 0-framed*)
Z = DisjUnion@{Rp[7, 1], Rp[2, 8], Rp[9, 3],
    Rp[4, 10], Rp[11, 5], ur[6], Km[12], Km[13], Km[14], Km[15], Km[16]};
For[i = 2, i ≤ 16, i++, Z = Z // m[1, i, 1]];
Z // Expand // Together
```

$$\left\{\frac{1-t+t^2-t^3+t^4}{t^2}, 0, 0, \frac{1}{\left(1-t+t^2-t^3+t^4\right)^2}\left(-t+2t^2-4t^3+6t^4-8t^5+8t^6-7t^7+4t^8+4c_1-6tc_1+6t^2c_1-4t^3c_1+4t^5c_1-6t^6c_1+6t^7c_1-4t^8c_1-4a_1w_1+2ta_1w_1-4t^2a_1w_1-4t^5a_1w_1+2t^6a_1w_1-4t^7a_1w_1\right)\right\}$$